

## EXCEL – NAME DES BLATTREGISTERS AUTOMATISCH IN ZELLE ÜBERNEHMEN

Bei der Arbeit mit mehreren Tabellenblättern gibt man diesen aussagekräftige Namen (siehe Beschriftung der Blattregister in Abbildung 1). Gleichzeitig enthalten die Tabellenblätter oft Überschriften (siehe Zelle B2 in Abbildung 1).

Kann man die Beschriftung der Register automatisch in eine Zelle im Arbeitsblatt übernehmen? Meistens wird diese Aufgabe in aufwändiger Handarbeit erledigt. Eine Funktion dafür gibt es nicht. Aber wie so oft in Excel kann man sich hier mit einem kleinen Umweg helfen.

	A	B	C	D	E	F	G	H	I	J	K	
1												
2		Absatz im Monat Februar in Stück										
3	Formel in B2:	="Absatz im Monat "&RECHTS(ZELLE("dateiname";A1);LÄNGE(ZELLE("dateiname";A1))-FINDEN("";ZELLE("dateiname";A1)))&" in Stück"										
4												
5												
6		Februar										
7	Formel in B6:	=RECHTS(ZELLE("dateiname";A1);LÄNGE(ZELLE("dateiname";A1))-FINDEN("";ZELLE("dateiname";A1)))										
8												
9												
10		C:\Users\...\Desktop\[Blattname.xlsx]Februar										
11	Formel in B10:	=ZELLE("dateiname";A1)										
12												
13												
14		45										
15	Formel in B14:	=LÄNGE(ZELLE("dateiname";A1))										
16												
17												
18		38										
19	Formel in B18:	=FINDEN("";ZELLE("dateiname";A1))										
20												
21												
22												
23												
24												
25												
26												

Abbildung 1

Man kann hierzu verschiedene Textfunktionen verwenden. Ich zeige die Lösung mit der Funktion **=RECHTS(Text; [Anzahl\_Zeichen])**. Die Funktion wird in Zelle B2 mit weiterem Text verkettet (manuell mit „&“ oder mit der **VERKETTEN**-Funktion). Nur der „Februar“ wird aus dem Blattregister übernommen. Verändert man jetzt die Beschriftung im Register, wird diese Veränderung in der Überschrift übernommen.

### VORÜBERLEGUNGEN

**Wichtig:** Die Arbeitsmappe muss gespeichert sein, sonst gibt es keinen Dateipfad.

Zunächst müssen wir eine Eigenschaft einer Zelle des Tabellenblattes auslesen. Dazu nutzen wir die Funktion **=ZELLE(Infotyp; [Bezug])**. Unter **Infotyp** kann man aus verschiedenen Eigenschaften der Zelle auswählen, die man mit dieser Funktion auslesen kann. **Bezug** gibt einen Zellbezug wieder, welchen ist in unserem Beispiel egal. In der Abbildung ist dies A1. Für uns ist der Dateipfad interessant.

Diesen können wir mit dem Infotyp „dateiname“ auslesen. In Abbildung 1 Zelle B10 sieht man das Ergebnis, den Pfad der Zelle A1. Wie lang der Dateipfad ist, hängt natürlich davon ab, wo ihre Arbeitsmappe gespeichert wurde. Der letzte Eintrag des Pfades der Zelle ist immer der Name des Tabellenblattes, in unserem Beispiel „Februar“. Unmittelbar davor steht aber immer der Name der Arbeitsmappe in eckigen Klammern, im Beispiel „[Blattname.xlsx]“.

Für uns ist die Position der schließenden eckigen Klammer interessant. Es ist immer das letzte Zeichen vor dem Namen des Tabellenblattes, welchen wir am Ende auslesen wollen. Die Position des Zeichens lesen wir mit der Funktion `=FINDEN(Suchtext; Text; [erstes_Zeichen])` aus (siehe Zelle B18 in Abbildung 1). Suchtext ist die Klammer, deren Position wir suchen. Text ist der Dateipfad, den wir mit dem Befehl `=ZELLE()` ermitteln. Das erste Zeichen benötigen wir nicht. Das Ergebnis gibt die Stelle der schließenden eckigen Klammer im Text als Zahl wieder.

Als letzte vorbereitende Überlegung benötigen wir jetzt noch die Länge des Dateipfades, die Anzahl der Zeichen. Dieses Problem lösen wir mit der Funktion `=LÄNGE(Text)`. Der Text ist in unserem Beispiel der mit `ZELLE` ausgelesene Dateipfad (siehe Abbildung 1 Zelle B14).

## LÖSUNG

Nach diesen Vorüberlegungen geht es in Abbildung 1 Zelle B6 ans Zusammenbauen:

`=RECHTS(Text; [Anzahl_Zeichen])`, das heißt, wir lesen aus einem Text eine bestimmte Anzahl von Zeichen von rechts aus gezählt aus. Da der Dateipfad und auch der Name des Tabellenblattes immer unterschiedlich lang sein können, müssen wir einige kleine Umwege gehen.

**Text:** Um an den Namen des Tabellenblattes zu kommen, lesen wir den Dateipfad einer Zelle mit `ZELLE` aus:

```
=RECHTS(ZELLE(„Dateiname“;A1);[Anzahl_Zeichen])
```

**[Anzahl\_Zeichen]:** Die Anzahl ist immer unterschiedlich, je nachdem wie lang der Name des Arbeitsblattes ist. Da der Name des Tabellenblattes immer der letzte Teil des Dateipfades nach der schließenden eckigen Klammer ist, bietet sich die Lösung an: Länge des Textes in der Zelle minus Position der schließenden eckigen Klammer. Das ergibt die Anzahl der von rechts auszulesenden Zeichen.

```
LÄNGE(ZELLE("dateiname";A1))-FINDEN("]";ZELLE("dateiname";A1))
```

Zusammen ergibt das (Abbildung 1, Zelle B6):

```
=RECHTS(ZELLE("dateiname";A1);LÄNGE(ZELLE("dateiname";A1))-FINDEN("]";ZELLE("dateiname";A1)))
```

In Abbildung 1, Zelle B2 habe ich diese Formel mit absoluten Textbestandteilen verbunden und damit die Überschrift erzeugt, in welcher der Monatsname dynamisch ist und sich der Beschriftung des Tabellenblattes anpasst.